/PROG  PNS0058
/ATTR
OWNER		= MNEDITOR;
COMMENT		= "53322606";
PROG_SIZE	= 3464;
CREATE		= DATE 15-08-11  TIME 19:35:36;
MODIFIED	= DATE 25-02-25  TIME 21:50:32;
FILE_NAME	= PNS0019;
VERSION		= 0;
LINE_COUNT	= 120;
MEMORY_SIZE	= 3912;
PROTECT		= READ_WRITE;
TCD:  STACK_SIZE	= 0,
      TASK_PRIORITY	= 50,
      TIME_SLICE	= 0,
      BUSY_LAMP_OFF	= 0,
      ABORT_REQUEST	= 0,
      PAUSE_REQUEST	= 0;
DEFAULT_GROUP	= 1,*,*,*,*;
CONTROL_CODE	= 00000000 00000000;
/APPL
/MN
   1:  !******************************** ;
   2:  !     COSMA METAL FORMING  ;
   3:  !******************************** ;
   4:  !  LINE: CHINA CHANGSHA ;
   5:  !  ROBOT: R12 ;
   6:  !  PROGRAM: PNS0058 ;
   7:  !  PART:53322606 ;
   8:  !ZYL15-8-11********************** ;
   9:   ;
  10:  !INITIALIZE ;
  11:  MOVE TO HOME    ;
  12:  R[14:CG SENSETIVE]=90    ;
  13:  R[18:PICK TIME BEFORE]=1    ;
  14:  R[19:DROP TIME BEFORE]=.1    ;
  15:   ;
  16:  !------------------------------- ;
  17:  !BEGINNING OF PROGRAM ;
  18:  CALL PARTDROP    ;
  19:  COL GUARD ADJUST R[14] ;
  20:  UTOOL_NUM=1 ;
  21:  UFRAME_NUM=0 ;
  22:  PAYLOAD[3] ;
  23:  R[5:SPEED]=10    ;
  24:  OVERRIDE=R[5:SPEED] ;
  25:  !INITIATE TO PICK-POUNCE ;
  26:J P[1:MIDDLE] 50% CNT100 ACC100    ;
  27:J P[2:PICK POUNCE] 50% CNT100 ACC100    ;
  28:  PAUSE ;
  29:  R[1:CYCLE TOTAL]=0    ;
  30:  TIMER[1]=RESET ;
  31:  TIMER[1]=START ;
  32:   ;
  33:  LBL[10:TOP OF LOOF] ;
  34:  OVERRIDE=R[5:SPEED] ;
  35:   ;
  36:  TIMER[3]=STOP ;
  37:  TIMER[2]=STOP ;
  38:  R[1:CYCLE TOTAL]=R[1:CYCLE TOTAL]+1    ;
  39:  R[2:LAST CYCLE TIME]=TIMER[2]    ;
  40:  R[3:LAST SPM]=60/R[2:LAST CYCLE TIME]    ;
  41:  R[8:R MOVE TIME]=TIMER[3]    ;
  42:  TIMER[2]=RESET ;
  43:  TIMER[2]=START ;
  44:  TIMER[3]=RESET ;
  45:   ;
  46:  !PROCEED TO PICK ;
  47:  LBL[20] ;
  48:  IF DI[1:P1 PICK ENTRY OK]=ON AND DI[3:FROM RL1 CLEAR]=ON AND DI[8:P1 PICK START]=ON,JMP LBL[25] ;
  49:  IF DI[29:CYCLE STOP]=ON,JMP LBL[90] ;
  50:  JMP LBL[20] ;
  51:   ;
  52:  LBL[25] ;
  53:  R[5:SPEED]=100    ;
  54:  TIMER[3]=START ;
  55:  OVERRIDE=R[5:SPEED] ;
  56:J P[3:PICK APPROACH] 90% CNT30 ACC100    ;
  57:   ;
  58:J P[4:PICK] 70% CNT0 TB R[18]sec,CALL PARTPICK ACC100 BREAK    ;
  59:  PAYLOAD[4] ;
  60:  CALL CHK_PICK    ;
  61:  MONITOR Z_CHECH ;
  62:L P[5:PICK DEPART] 2500mm/sec CNT30 ACC100 TA R[20]sec,DO[5:TO RL1 DROP START]=PULSE,1.0sec    ;
  63:  DO[8:P1 PARTS OUT]=PULSE,1.0sec ;
  64:  LBL[27] ;
  65:  IF DI[9:VACUUM ON L]=OFF OR DI[11:VACUUM ON R]=OFF,JMP LBL[27] ;
  66:   ;
  67:L P[6:PICK SNEAK] 2500mm/sec CNT100 ACC100    ;
  68:J P[7:PICK CLEAR] 80% CNT100 ACC100    ;
  69:   ;
  70:  LBL[30:GOTO DROP POUNCE] ;
  71:J P[8:DROP POUNCE] 100% CNT100 ACC100 BREAK    ;
  72:   ;
  73:  TIMER[3]=STOP ;
  74:  LBL[40:AT DROP POUNCE] ;
  75:  IF DI[2:P2 DROP ENTRY]=ON AND DI[4:FROM R23 CLEAR]=ON AND DI[5:P2 DROP START]=ON,JMP LBL[80] ;
  76:  IF DI[7:PRESS STAMPING COMPLETE]=ON AND DI[2:P2 DROP ENTRY]=ON,JMP LBL[50] ;
  77:  IF DI[18:MOVE TO CLEAR POS]=ON,JMP LBL[60] ;
  78:  JMP LBL[40] ;
  79:   ;
  80:  LBL[50:MOVE DROP SNEAK] ;
  81:  TIMER[3]=START ;
  82:  OVERRIDE=R[5:SPEED] ;
  83:J P[9:DROP SNEAK] 90% CNT100 ACC100 BREAK    ;
  84:  LBL[51:DROP OR C-P-S] ;
  85:  IF DI[2:P2 DROP ENTRY]=ON AND DI[4:FROM R23 CLEAR]=ON AND DI[5:P2 DROP START]=ON,JMP LBL[80] ;
  86:  IF DI[18:MOVE TO CLEAR POS]=ON,JMP LBL[40] ;
  87:  JMP LBL[51] ;
  88:  JMP LBL[50] ;
  89:   ;
  90:  LBL[60:MOVE CLS-PRS-ACC] ;
  91:J P[1:MIDDLE] 50% CNT100 ACC100    ;
  92:  PAUSE ;
  93:  JMP LBL[30] ;
  94:   ;
  95:  LBL[80:DROP ENABLED] ;
  96:  TIMER[3]=START ;
  97:  !PROCEED TO DROP ;
  98:  OVERRIDE=R[5:SPEED] ;
  99:J P[14:DROP LIMIT] 90% CNT100    ;
 100:L P[10:DROP APPROACH] 3000mm/sec CNT40 ACC100    ;
 101:  MONITOR END Z_CHECH ;
 102:J P[11:DROP] 90% CNT5 TB R[19]sec,CALL PARTDROP ACC100    ;
 103:  DO[6:PART DROP COMPLETE]=PULSE,1.0sec ;
 104:  PAYLOAD[3] ;
 105:  CALL CHK_DROP    ;
 106:   ;
 107:J P[12:DROP DEPART] 100% CNT50 ACC110 TA    .20sec,DO[7:PRESS STAMPING START]=PULSE,1.0sec    ;
 108:J P[13:DROP CLEAR] 100% CNT100 ACC110    ;
 109:  !GOTO PICK POUNCE ;
 110:J P[2:PICK POUNCE] 100% CNT100 ACC100 BREAK    ;
 111:  IF DI[29:CYCLE STOP]=ON,JMP LBL[90] ;
 112:  JMP LBL[10] ;
 113:   ;
 114:  LBL[90:END OF CYCLE] ;
 115:  TIMER[1]=STOP ;
 116:  R[4:TOTAL CYCLE TIME]=TIMER[1]    ;
 117:J P[1:MIDDLE] 50% CNT100    ;
 118:  MOVE TO HOME    ;
 119:   ;
 120:  LBL[100:END OF PROGRAM] ;
/POS
P[1:"MIDDLE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2086.456  mm,	Y =     -.007  mm,	Z =  -960.000  mm,
	W =  -180.000 deg,	P =     -.000 deg,	R =  -180.000 deg,
	E1=   107.700 deg
};
P[2:"PICK POUNCE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2607.002  mm,	Y =  1540.281  mm,	Z =  -953.798  mm,
	W =  -180.000 deg,	P =     -.000 deg,	R =   119.876 deg,
	E1=    72.613 deg
};
P[3:"PICK APPROACH"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2673.511  mm,	Y =  3293.913  mm,	Z =  -884.380  mm,
	W =  -179.974 deg,	P =      .022 deg,	R =    89.723 deg,
	E1=   106.895 deg
};
P[4:"PICK"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2673.683  mm,	Y =  3294.041  mm,	Z =  -998.620  mm,
	W =  -179.974 deg,	P =      .022 deg,	R =    89.723 deg,
	E1=   106.895 deg
};
P[5:"PICK DEPART"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2751.041  mm,	Y =  3293.958  mm,	Z =  -780.628  mm,
	W =  -179.972 deg,	P =      .024 deg,	R =    89.723 deg,
	E1=   106.895 deg
};
P[6:"PICK SNEAK"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2799.105  mm,	Y =  2108.561  mm,	Z =  -780.499  mm,
	W =  -179.974 deg,	P =      .023 deg,	R =    89.723 deg,
	E1=   106.895 deg
};
P[7:"PICK CLEAR"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 0',
	X =  2065.615  mm,	Y =  1511.702  mm,	Z =  -830.937  mm,
	W =   179.996 deg,	P =     -.022 deg,	R =   132.298 deg,
	E1=   112.741 deg
};
P[8:"DROP POUNCE"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2028.780  mm,	Y =  -832.522  mm,	Z =  -944.863  mm,
	W =   179.949 deg,	P =      .081 deg,	R =  -145.959 deg,
	E1=   123.964 deg
};
P[9:"DROP SNEAK"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2659.427  mm,	Y = -1613.796  mm,	Z =  -998.093  mm,
	W =  -179.975 deg,	P =      .129 deg,	R =  -110.979 deg,
	E1=   128.766 deg
};
P[10:"DROP APPROACH"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2610.199  mm,	Y = -3271.375  mm,	Z =  -962.000  mm,
	W =  -179.976 deg,	P =      .131 deg,	R =   -89.772 deg,
	E1=   106.327 deg
};
P[11:"DROP"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2610.199  mm,	Y = -3271.375  mm,	Z = -1164.148  mm,
	W =  -179.976 deg,	P =      .131 deg,	R =   -89.772 deg,
	E1=   106.327 deg
};
P[12:"DROP DEPART"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2610.199  mm,	Y = -3271.375  mm,	Z = -1109.000  mm,
	W =  -179.976 deg,	P =      .131 deg,	R =   -89.772 deg,
	E1=   106.327 deg
};
P[13:"DROP CLEAR"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2369.893  mm,	Y = -1101.843  mm,	Z = -1048.080  mm,
	W =  -179.799 deg,	P =     -.436 deg,	R =  -140.127 deg,
	E1=   105.955 deg
};
P[14:"DROP LIMIT"]{
   GP1:
	UF : 0, UT : 1,		CONFIG : 'N U T, 0, 0, 1',
	X =  2659.427  mm,	Y = -1613.796  mm,	Z =  -998.093  mm,
	W =  -179.975 deg,	P =      .129 deg,	R =  -110.979 deg,
	E1=   128.766 deg
};
/END
